Jede verschickte E-Mail automatisch per CC/BCC verteilen, Teil 1
Versionen: Outlook 2003, 2002/XP und 2000

In einer früheren Ausgabe von Outlook Weekly hatten wir Ihnen eine VBA-Lösung vorgestellt, mit der Sie von jeder erstellten E-Mail automatisch eine sichtbare (CC) oder unsichtbare (BCC) Kopie versschicken können. Die Resonanz auf diesen Beitrag war sehr groß und uns haben zahlreiche Zuschriften von Lesern erreicht, die diese Funktion gerne um eine Steuerung "CC senden Ja/Nein" beziehungsweise "BCC Senden Ja/Nein" erweitern würden. Wir haben eine entsprechende Änderung in die Lösung integriert, die wir Ihnen mit diesem dreiteiligen Tipp der Woche vorstellen möchten.

Zunächst noch einmal kurz zum Konzept der Lösung: Um von jeder gesendeten Nachricht eine sichtbare (CC) oder unsichtbare (BCC) Kopie zur Kontrolle oder zu Dokumentationszwecken an eine bestimmte E-Mail-Adresse zu versenden, müssen Sie bei jeder neuen Nachricht die Felder "CC" bzw. "BCC" füllen. Abgesehen davon, dass dieses Vorgehen bei jeder Nachricht zusätzlichen Aufwand bedeutet, besteht die Gefahr, dass das Setzen des CC- oder BCC-Feldes in der Alltagshektik vergessen wird. Outlook bietet jedoch Möglichkeiten, diesen Vorgang zu automatisieren: Bei jeder verschickten E-Mail wird ein Anwendungsereignis "ItemSend" ausgelöst, das Sie um eigene Aktionen ergänzen können. Eine solche Aktion kann dann zum Beispiel auch das "CC"- oder "BCC"-Feld automatisch setzen. Dabei ist zu berücksichtigen, dass eventuell nicht jede Nachricht in CC oder BCC versendet werden soll. Um diesem Umstand Rechnung zu tragen, könnte man die Lösung um zwei Abfragen "CC versenden?" bzw. "BCC versenden?" ergänzen, die dann jeweils mit einem Klick auf Ja oder Nein zu beantworten sind. Wenn Sie dann jedoch mehrere Nachrichten hintereinander ohne CC oder BCC versenden möchten, müssten Sie die Abfragen entsprechend oft bestätigen. Um diesen Aufwand zu vermeiden, haben wir uns entschlossen, eine Symbolleiste mit zwei Schaltflächen CC: an/aus und BCC: an/aus einzusetzen: Ein Klick schaltet die jeweilige Funktion aus, ein weiterer Klick schaltet sie wieder ein. Vor dem Versenden einer Nachricht stellen Sie dann einfach den CC- und BCC-Status über diese Schaltflächen entsprechend ein und klicken dann auf Senden.

Um diese solche Lösung in Ihr Outlook einzubauen, gehen Sie wie folgt vor:

1. Wechseln Sie mit Alt+F11 in die VBA-Entwicklungsumgebung.

2. Öffnen Sie den Bereich "VBAProject|Microsoft Outlook Objekte|DieseOutlooksitzung" per Doppelklick.

3. Geben Sie im allgemeinen Teil die folgenden Deklarationen ein:

Option Explicit

Const cstrBCCAdr = "joeuser@provider.de"
Const cstrCCAdr = "chef@provider.de"

Dim CmdBar As Office.CommandBar

Dim SendCC As Boolean
Dim SendBCC As Boolean


Die Konstanten "cstrCCAdr" und "cstrBCCAdr" passen Sie dabei entsprechend an. Soll generell keine CC- oder BCC-Kopie gesendet werden, setzen Sie die jeweilige Konstante auf eine leere Zeichenkette:

  Const cstrBCCAdr = ""
  Const cstrCCAdr = "chef@provider.de"

oder

  Const cstrBCCAdr = "joeuser@provider.de"
  Const cstrCCAdr = ""
· Geben Sie danach die folgenden Prozeduren ein:

Public Sub SwitchCC()

  SendCC = Not SendCC
  UpdateCmdBar

End Sub

Public Sub SwitchBCC()

  SendBCC = Not SendBCC
  UpdateCmdBar

End Sub

Sub UpdateCmdBar()

  On Error Resume Next
  If CmdBar Is Nothing Then
    Beep
    MsgBox "Symbolleiste 'CCBCC' nicht gefunden...", _
           vbOKOnly + vbCritical, "!!! Problem !!!"
    Exit Sub
  End If

  With CmdBar
    .Controls(1).Caption = IIf(SendCC, _
                              "CC: an", "CC: aus")

    .Controls(2).Caption = IIf(SendBCC, _
                               "BCC: an", "BCC: aus")

  End With

End Sub

Die Prozeduren "SwitchCC" und "SwitchBCC" schalten den aktuellen Status der globalen Variablen "SendCC" und "SendBCC" jeweils um, ändern den Status also per Operator "Not" von "True" auf "False" beziehungsweise umgekehrt. Die Prozedur "UpdateCmdBar" sorgt dafür, dass der aktuelle Status "an" oder "aus" in der Symbolleiste entsprechend angezeigt wird.

· Speichern Sie die Änderungen und verlassen Sie die VBA-Entwicklungsumgebung wieder.

Soviel erst einmal für diesen Teil des Tipps der Woche. Im zweiten Teil schaffen wir die Basis für das Handling der Anwendungsereignisse.

Im ersten Teil dieses Tipps der Woche haben wir zunächst die Hilfsprozeduren eingegeben. Dieser Teil ergänzt die Lösung nun um die Ereignisprozedur für das Handling der Anwendungsereignisse:

1. Wechseln Sie mit Alt+F11 in die VBA-Entwicklungsumgebung.

2. Öffnen Sie den Bereich "VBAProject|Microsoft Outlook Objekte|DieseOutlooksitzung" per Doppelklick.

3. Stellen Sie im VBA-Editor die Liste Objekt (links) auf "Application" und die Liste Prozedur (rechts) auf "ItemSend" ein.

4. Geben Sie die folgenden Anweisungen ein:

Private Sub Application_ItemSend(ByVal Item As Object, _
                                 Cancel As Boolean)
  Dim objMail As Outlook.MailItem
  Dim objRecpt As Outlook.Recipient

  On Error Resume Next
  Exit Sub '!!!!!
  If TypeName(Item) = "MailItem" Then 'nur bei E-Mails...
    On Error Resume Next
    Set objMail = Item
    With objMail
      If SendCC And cstrCCAdr <> "" Then
        If .Recipients.Item(cstrCCAdr) Is Nothing Then
          Set objRecpt = .Recipients.Add(cstrCCAdr)
          objRecpt.Type = olCC
          .Recipients.ResolveAll
        End If
      End If
      If SendBCC And cstrBCCAdr <> "" Then
        If .Recipients.Item(cstrBCCAdr) Is Nothing Then
          Set objRecpt = .Recipients.Add(cstrBCCAdr)
          objRecpt.Type = olBCC
          .Recipients.ResolveAll
        End If
      End If
    End With
    Set objRecpt = Nothing
    Set objMail = Nothing
  End If

End Sub

In Zukunft werden nun bei jedem Klick auf Senden durch das damit ausgelöste Anwendungsereignis "ItemSend" die oben gezeigten Anweisungen ausgeführt. Diese prüfen zunächst, ob es sich bei dem zu sendenden Element um eine Nachricht (MailItem) handelt. Wenn ja, erfolgt eine Prüfung der globalen Variablen "SendCC" und "SendBCC" beziehungsweise der Konstanten "cstrCCAdr" und "cstrBCCAdr" - ein Versand in CC oder BCC erfolgt dann nur, wenn die Option eingeschaltet ist und die Konstante einen Inhalt hat. Danach wird jeweils geprüft, ob die angegebene CC- oder BCC-Adresse bereits gesetzt ist. Das ist notwendig, damit beispielsweise nach einer Änderung der Nachricht im Postausgang die CC- beziehungsweise BCC-Adresse nicht ein zweites, drittes, viertes Mal hinzugefügt wird. Sind CC- beziehungsweise BCC-Adresse noch nicht gesetzt, so werden sie per "Recipients.Add" zur Auflistung der Empfänger hinzugefügt. Über die Eigenschaft "Type" legen wir abschließend fest, ob es sich beim Adressaten um den Empfänger einer CC-Kopie (olCC) oder einer BCC-Kopie (olBCC) handelt.

Bitte beachten Sie, dass bei einzelnen Outlook-Versionen durch "Recipients.Add" die Sicherheitsabfrage ausgelöst wird, die mit Aktivierung der Option Zugriff gewähren für 1 Minute und einem Klick auf die Schaltfläche Ja zu beantworten ist.

Stellen Sie nun im VBA-Editor die Liste Prozedur (rechts) auf "Startup" ein und geben Sie die folgenden Anweisungen ein:

Private Sub Application_Startup()

  On Error Resume Next
  Exit Sub '!!!!!
  'Default: CC und BCC senden
  SendCC = True
  SendBCC = True
  'Referenz auf Symbolleiste 'CCBCC'
  Set CmdBar = ActiveExplorer.CommandBars("CCBCC")
  If CmdBar Is Nothing Then
    Beep
    MsgBox "Symbolleiste 'CCBCC' nicht gefunden...", _
           vbOKOnly + vbCritical, "!!! Problem !!!"
    SendCC = False
    SendBCC = False
  Else
    UpdateCmdBar
  End If

End Sub

Diese Anweisungen setzen beim Start von Outlook die globalen Variablen "SendCC" und "SendBCC" auf "True" und legen so fest, dass eine CC- beziehungsweise eine BCC-Kopie gesendet werden soll. Wenn Sie häufig CC- aber nur nur gelegenlich BCC-Kopien oder umgekehrt versenden, ändern Sie die Zuweisungen einfach auf beispielsweise "SendBCC = False" ab. Danach wird eine globale Referenz auf die Symbolleiste "CCBCC" gesetzt, um Änderungen einfacher durchreichen zu können. Wird die Symbolleiste nicht gefunden, erfolgt die Anzeige eines entsprechenden Hinweises, andernfalls sorgt der Aufruf von "UpdateCmdBar" dafür, dass der aktuelle Zustand "an" beziehungsweise "aus" für die beiden Schaltfächen aktualisiert und sichtbar gemacht wird.

Stellen Sie im VBA-Editor dann die Liste Prozedur (rechts) auf "Quit" ein und geben Sie die folgenden Anweisungen ein:

Private Sub Application_Quit()

  Set CmdBar = Nothing

End Sub

Diese Anweisung sorgt dafür, dass die globale Objektvariable für den Zugriff auf die Symbolleiste "CCBCC" wieder zurückgesetzt und der damit verbundene Speicherbereich wieder freigegeben wird. Speichern Sie die Änderungen schließlich und verlassen Sie die VBA-Entwicklungsumgebung wieder.

Die Lösung ist nun bezüglich des Handlings der Anwendungsereignisse einsatzbereit, es fehlt jedoch noch die Symbolleiste mit den Schaltflächen zum Umschalten des Status. Wie sie in Outlook angelegt und eingerichtet wird, ist Inhalt des letzten Teils dieses Tipps. Damit Ihnen die Lösung bis dahin nicht bei jedem Start und bei jedem Senden Fehlermeldungen wegen der noch fehlenden Symbolleiste anzeigt, haben wir die eigentliche Ausführung mit entsprechenden "Exit Sub"-Anweisungen in "Application.Startup" und "Application.SendItem" erst einmal "lahm gelegt" - diese Anweisungen sind nach dem Anlegen der Symbolleiste zu löschen.

Nachdem wir im ersten Teil dieses Tipps der Woche die Basis in Form von Ereignisprozedur für die Anwendungsereignisse und Hilfsprozeduren für das Handling der Symbolleiste geschaffen haben, wollen wir nun die Symbolleiste "CCBCC" in Outlook anlegen und einrichten.

1. Klicken Sie mit der rechten Maustaste auf das Menü oder eine Symbolleiste und wählen Sie Anpassen im Kontextmenü an.

2. Klicken Sie auf der Registerkarte "Symbolleisten" auf die Schaltfläche Neu, benennen Sie die Symbolleiste "CCBCC" und klicken Sie auf Ok.

3. Wechseln Sie auf die Registerkarte "Befehle" und markieren Sie den Eintrag "Makros" in der Liste Kategorien.

4. Ziehen Sie die Makros "SwitchCC" und "SwitchBCC" aus der Liste Befehle in die neue Symbolleiste "CCBCC".

5. Klicken Sie die erste Schaltfläche (für CC) mit der rechten Maustaste an und geben Sie im Feld Name die Zeichenfolge "CC: an" ein.

6. Klicken Sie nochmals die erste Schaltfläche mit der rechten Maustaste an, wählen Sie Eigenschaften im Kontextmenü an und aktivieren Sie die Option Nur Text (Immer).

7. Klicken Sie die zweite Schaltfläche (für BCC) mit der rechten Maustaste an und geben Sie im Feld Name die Zeichenfolge "BCC: an" ein.

8. Klicken Sie erneut die zweite Schaltfläche mit der rechten Maustaste an, wählen Sie das Kontextmenü Eigenschaften an und aktivieren Sie die Option Nur Text (Immer).

9. Klicken Sie nochmals die zweite Schaltfläche mit der rechten Maustaste an, wählen Sie Eigenschaften im Kontextmenü an und aktivieren Sie die Option Gruppierung beginnen.

10. Klicken Sie im Dialog "Anpassen" auf die Schaltfläche Schließen.

Die Symbolleiste ist damit einsatzbereit. Bei einem Klick auf eine der Schaltflächen wird die dazugehörige Prozedur "SwitchCC" bzw. "SwitchBCC" aufgerufen. Diese Prozeduren schalten den aktuellen Status in den globalen Variablen "SendCC" und "SendBCC" um und rufen dann "UpdateCmdBar" auf, damit der aktuelle Status in den Schaltflächen der Symbolleiste wiedergegeben wird.

Die eigentliche Ausführung der Lösung hatten wir zunächst mit entsprechenden "Exit Sub"-Anweisungen in "Application.Startup" und "Application.SendItem" erst einmal "lahm gelegt" - diese Anweisungen sind nun nach dem Anlegen der Symbolleiste zu löschen:

1. Wechseln Sie mit Alt+F11 in die VBA-Entwicklungsumgebung.

2. Öffnen Sie den Bereich "VBAProject|Microsoft Outlook Objekte|DieseOutlooksitzung" per Doppelklick.

3. Stellen Sie im VBA-Editor die Liste Objekt (links) auf "Application" und die Liste Prozedur (rechts) auf "ItemSend" beziehungsweise "Startup" ein.

4. Löschen Sie jeweils die Zeilen

Exit Sub '!!!!!
· Speichern Sie die Änderungen und verlassen Sie die VBA-Entwicklungsumgebung wieder.

Die Lösung ist jetzt komplett einsatzbereit. Wenn Sie auf eine der Schaltfläche in der Symbolleiste klicken, sehen Sie, wie der Status entsprechend umgeschaltet wird. Abhängig vom jeweils eingestellten Status werden Ihre Nachrichten dann automatisch an die festgelegte CC- beziehungsweise BCC-Adresse versendet.

